APPENDIX

Sample `SAS PROC NLMIXED` syntax

Below is sample syntax for the 3-level mixed-effects location scale (MELS) model used in this paper. Expressions with all uppercase letters denote SAS-specific syntax, whereas expressions including lowercase letters are for user-defined entities. The dependent variable `NAchange` is the change in negative affect associated with a smoking event (now-before), and, for simplicity, we only consider the covariates `wave` (coded 0=time1, 1=time2) and `alc` (coded 0=no alcohol use, 1=alcohol use). The variable `id` is a subject-level identifier.

The first part of the syntax is for reading in the data, creation of wave indicator variables, and creation of the between-subjects and within-subjects components of the occasion-varying variable `alc`. The former is the subject mean of this variable, and the latter is the occasion-specific deviation of the variable relative to the subject mean.

```
FILENAME in1 'c:\NAchange.dat';
DATA one; INFILE in1;
INPUT id wave NAchange alc;

/* create indicators of the two waves */
in1=0; in2=0;
IF wave=0 THEN in1=1;
IF wave=1 THEN in2=1;

/* create the subject mean of the alc variable */
PROC SORT; BY id;
PROC MEANS NOPRINT; CLASS id; VAR alc;
OUTPUT OUT = oneb MEAN = bs_alc;

/* merge the mean data with the original data */
/* create the occasion deviation of alc relative to the mean */
DATA onec; MERGE one onec; BY id;
ws_alc = alc - bs_alc;
```

In the NLMIXED code below, the mean response model is given by `mean`, with regression coefficients named `b_0, b_wave, b_bs_alc`, and `b_ws_alc`. For the random subject effects, upsilon is the random location and omega is the random scale effect. The model for the within-subjects (error) variance is denoted vare, with e_0 for the reference variance (i.e., the variance when all covariates equal 0), and regression coefficients named e_wave, e_bs_alc, and e_ws_alc. Finally, v_upsilon, v_wave, and v_omega represent the variances of the subject location, wave, and subject scale random effects, respectively, with subject location scale covariance given by cov.

The syntax below uses PROC NLMIXED as a 2-level model, but with a "trick" described by Dale McLerran of the Fred Hutchinson Cancer Research Center in an old web post, which has since been deleted, to estimate a 3-level model (i.e., observations nested within waves nested within subjects). For the current example, we created indicator variables for the two measurement waves (in the syntax above), in1 and in2. These are then included in the mean model and specified as random effects (named below as w1 and w2) with mean zero. Furthermore, they are constrained to have the same variance (v_wave), and to be independent of each other and the subject random effects.

```
/* 3-level MELS model – Negative Affect */
PROC NLMIXED GCONV=1e-12;
PARMS b_0=0.46 b_wave=-0.06 b_bs_alc=-0.27 b_ws_alc=-0.09
      e_0=0.20 e_wave= 0.16 e_bs_alc=-0.20 e_ws_alc= 0.19
       v_upsilon=0.5 v_omega=0.005 v_wave=0.1 cov=0;
mean = b_0 + b_wave*wave + b_bs_alc*bs_alc + b_ws_alc*ws_alc +
       + upsilon + in1*w1 + in2*w2;
vare = EXP(e_0 + e_wave*wave + e_bs_alc*bs_alc + e_ws_alc*ws_alc +
omega);
MODEL NAchange ~ NORMAL(mean,vare);
RANDOM upsilon omega w1 w2 ~ NORMAL([0,0,0,0],
       [v_upsilon, cov,v_omega, 0,0,v_wave, 0,0,0,v_wave]) SUBJECT=id;
RUN;
```

Users must provide starting values for all parameters on the PARMS statement. To do so, it is beneficial to run the model in stages using estimates from a prior stage as starting values and setting the additional parameters to zero or some small value. For example, one can start by estimating a three-level random-intercept model using standard mixed model software (i.e., SAS PROC MIXED) to yield starting values for the fixed effects (β), subject variance (`v_upsilon`), wave variance (`v_wave`), and error variance (`e0`). Then, one can add covariates to the error variance model, perhaps one at a time, with starting values of zero. Finally, the full model with the parameters associated with the random scale effect (`v_omega, cov`) can be estimated. In practice, this approach works well with PROC NLMIXED, which sometimes has difficulties in converging to a solution for complex models. Also, in our experience, it seems that specifying a small starting value for the random scale variance (`v_omega`) helps model convergence. Furthermore, for complex models, it is sometimes the case that the default convergence criterion is not strict enough. In the above syntax, the convergence criterion is specified as `GCONV=1e-12` on the `PROC NLMIXED` statement.